


Please use the space provided on the exam for your answers to the following questions. Note that question values are shown in parentheses.

1. (18) Consider the following user interface that includes a list box named `lstFruit` (MultiSelect property set to 0-None) and two command buttons:



When the user clicks on the “Move Up” button, the word selected in the list box moves up one spot. In the case above, “Grape” would move to the second line and “Banana” would move to the third line (they would exchange locations).

Write the click event for the “Move Up” command button (`cmdMoveUp`). Be sure that the code does not cause an error if the selected item is already at the top.

2. (18) What output would be generated from the following code segment?

```
Dim n As Integer, j As Integer, k As Integer
n = 0
For j = 5 To 1 Step -1
    n = n + 1
    For k = n To j
        Form1.Print j, k
    Next k
Next j
```

3. (16) You have the following three arrays:

Price	Qty	Gross
1	1	1
2	2	2
3	3	3
4	4	4
5	5	5

Assume that the first cell of each array stores information on product 1, the second cell of each array stores information on product 2, etc.

Write a code segment (use page 3) that:

- Multiplies Price times Qty for each product and stores the result in the corresponding cell in Gross.
- Scans the values in the Gross array and reports (via a Form1.Print) the number of the product with the largest gross amount (assume there are no two cells with the same value).

Answer Question 3 below:

4. (14) Microsoft Access includes the ability to create forms with controls (text boxes, list boxes, etc.) just like Visual Basic. Access also supports its own version of Visual Basic (Visual Basic for Applications). If a company uses a database that is currently an Access database, explain the main advantage of using Visual Basic to create the user interface instead of Microsoft Access.

5. (16) Assume you have a string variable defined as follows:


```
Dim phrase As String  
phrase = "name no one man"
```

Write a code segment that


- a. Removes all blanks from the original string variable.
- b. Using the “blank-free” string, compares the first letter to the last letter, the second letter to the second to last letter, the third letter to the third from the last letter, etc.
- c. If the comparisons in step (b) above show that all the corresponding letters are the same, have a message box display the message “It’s a palindrome”.
- d. If any of the corresponding letters compared in step (b) above are not equal, have a message box display the message “It’s not a palindrome”.

The objective is to determine if the phrase is spelled the same way forward and backward. You may declare any additional variables as needed. Your solution should work for any sentence stored in the variable “phrase” assuming that only lowercase letters are used.

6. (18) Consider the following user interface:



Clicking on the search button generates the following message box:



The code for the “Search” command button is:

```
Private Sub cmdSearch_Click()  
    Dim originalPhrase As String, searchPhrase As String, count As Integer  
    originalPhrase = txtOrigPhrase.Text  
    searchPhrase = txtSubPhrase.Text  
    count = phraseCount(originalPhrase, searchPhrase)  
    MsgBox searchPhrase & " occurs " & count & " times."  
End Sub
```

Write the complete function “phraseCount” referred to in the code above.

Answer Key

1. One solution follows:

```
Private Sub cmdMoveUp_Click()  
    Dim n As Integer  
    Dim temp As String  
    With lstFruit  
        n = .ListIndex  
        If n > 0 Then  
            temp = .List(n)  
            .List(n) = .List(n - 1)  
            .List(n - 1) = temp  
            .ListIndex = n - 1  
        End If  
    End With  
End Sub
```

2. The values printed are:

j	k
5	1
5	2
5	3
5	4
5	5
4	2
4	3
4	4
3	3

3. A possible solution is:

```
For k = 1 To 5  
    Gross(k) = Price(k) * Qty(k)  
Next k  
max = Gross(1)  
loc = 1  
For k = 2 To 5  
    If Gross(k) > max Then  
        max = Gross(k)  
        loc = k  
    End If  
Next k  
Form1.Print "Maximum gross is product " & loc
```

4. The main advantage of using VB instead of Access is the independence provided by VB. VB can handle Access databases as well as other ODBC (Open Database Connectivity) databases. Thus, if a decision were made to change database technology and convert the database to Oracle or some other database, the VB program would still work. If the entire application were created using Access, a conversion to another database would mean the entire application would have to be rewritten.
5. A sample solution is:

```
Dim phrase As String
phrase = "name no one man"

Dim noBlanks As String, k As Integer, j As Integer
For k = 1 To Len(phrase)
    If Mid$(phrase, k, 1) <> " " Then
        noBlanks = noBlanks & Mid$(phrase, k, 1)
    End If
Next k
j = Len(noBlanks)
For k = 1 To Len(noBlanks) \ 2
    If Mid$(noBlanks, k, 1) <> Mid$(noBlanks, j, 1) Then
        MsgBox "I t's not a palindrome"
        Exit Sub
    End If
    j = j - 1
Next k
MsgBox "I t's a palindrome"
```

6. A sample solution follows:

```
Public Function phraseCount(ByVal phrase As String, _
                           ByVal subPhrase As String) As Integer
    Dim foundLoc As Integer
    Dim n As Integer
    foundLoc = InStr(1, phrase, subPhrase)
    n = 0
    Do While foundLoc > 0
        n = n + 1
        foundLoc = InStr(foundLoc + 1, phrase, subPhrase)
    Loop
    phraseCount = n
End Function
```